[
Redo all room descriptions

BETA 2 
=====
- Shortened into
- Moved portion of intro to a flashback on first turn to break up the lengthy reading
- Reworked many room descriptions
- Many changes to remove passive voice wordings
- Added starter map to the package
- Added walkthrough to the package
- added garage door functionality, driveway location, choice of freedom (running away when in driveway)
- Change knife cut package to 5 points
- Fixed giving points multiple times when re-opening the package or re-pushing the china cabinet
- Snoring clue (in Rec room) only once
- The Important Looking Switch on the machinery now recognizes PUSHING, PULLING, FLIPPING, na d SWITCHING ON as PRESSING.
- Updated help & about screens
- Moved the papers, pencils, and candy wrappers to the desk drawer.
- Changed the papers and candy wrappers to be takable objects to reduce frustrations about not being able to take/manipulate objects in the environment.
- Fixed papers description
- Added "Pencil" to be understood as "Pencils"
- Added Walkthrough command
- Removed the TV stating it was turned off twice
- Corrected "Search Mirror" while in the bedroom providing the response for the bathroom mirror.

RELEASE 1 - Oct 1, 2009
================

RELEASE 2 - Nov 16, 2009
================
Corrected "Richard Boss" to "Richard Bos" in beta testing credits
Fixed "Property Tage" typo that allowed creation of an unimplemented object in the secret lab
Changed "Keys" from "a keys" to "some keys"
Corrected remaining typeos and grammer issues in text
Implemented shabby railing correctly (was not placed on front porch correctly)
Fixed missing harness description
Fixed issue that suppressed what TV show was being shown at certain times for a switched on TV.
Added code to disallow taking of scenery objects during a GET ALL command


todo
====
Fix gramaticals based on beta test & reviews
Read all reviews for needed fixes
Implement Exterminate command
Add a Just for Fun? section
Cut Gilby when he's asleep, he easily avoids the cut as if he's awake
Add additional verbs for cleaning up the pool
How about cutting Gilby as a wake up method?


Questions:


After taking keys for the first time:
	Award 20 points;
	Continue the action
Before taking keys:
	if Gilby has the keys and Gilby is on the bed begin;
		say "You attempt to take the keys from Gilby, but as you tug at them, he begins to move. He snatches the keys away, eyes you wearily, and shuffles a bit in the bed."; 
		Now the grumpiness of Gilby is the grumpiness of Gilby plus 1;
		stop the action;
	otherwise; 
		continue the action;
	end if.

]

"Interface" by Ben Vegiard.
The story genre is "Science Fiction".
The story headline is "A different kind of out-of-body experience,".
The story creation year is 2009.
The release number is 2.
Use the serial comma. 

Include Patrollers by Michael Callaghan.
Include Basic Help Menu by Emily Short.
Include Locksmith by Emily Short.
[
Include Object Response Tests by Juhana Leinonen.
]
 Include Property Checking by Emily Short.


Section 1 - Setup

TVShowNumber is a number that varies.
Stereo_count is a number that varies.
AwakeDescrNumber is a number that varies.
SnoringClueGiven is a number that varies.
A thing can be _awarded or _unawarded.
A thing is usually _unawarded.

When play begins: 
Change the time of day to 7:47 AM;
Change the TVShowNumber to 1;
Change the SnoringClueGiven to 0;
Change AwakeDescrNumber  to 1;
Change the right hand status line to "[Score] / [time of day]";
Say "Uncle Floyd is your favorite person in the entire world, next to Mom and Dad, of course!  He is the inventor of several hi-tech gadgets that have been so successful that he started his own thriving electronics company.  Now he spends his time doing what he likes best: thinking about what could be and letting his staff geniuses work out the details.  He's always bustling with enthusiasm to show off his company's various inventions to you, long before the general public can see them.  Your favorite has to be the 'Fly on the Wall'; a remote controlled flying camera that is so small it appears to be an ordinary housefly unless you look at it very closely.";
say "[line break]";
say "Today you can barely contain your excitement since Uncle Floyd has invited you to his lab with the promise of showing you his company's best invention yet, even better than the Fly on the Wall.  He jovially refuses to tell you anything about the new project.";
say "[line break]";
Say "You race him down the hall where he stops to slip a security card into a slot mounted next to a solid metal door.  The security door gives way to a breathtaking room, the walls of which are lined with various blinking electronic components and colored wiring.  Centered prominently in the room is a throne-like chair with several odd-looking controls and LCD panels.";
say "[line break]";
Say "Another man, scrawny and wearing an overly thick and unkempt mustache, is already in the room as you arrive.  Your Uncle introduces him as Gilby, the 'make it happen' guy of the new device.  When you ask Uncle Floyd what the new device is, he simply answers, 'Have a seat and we'll SHOW you!  You get to be the first, outside of our testers, to try it out!'";
say "[line break]";
say "A little reluctantly, you sit in the chair.  Your Uncle and Gilby give each other a gleeful look and then begin to strap you into the seat using bindings you did not notice before.  To settle your growing protests, your Uncle remarks 'Ah!  But you will be the first!  It will all be worth it!  Just wait and see!'";
say "[line break]";
say "Suddenly, Gilby disappears behind the chair and you feel a hat being placed on your head followed by a heart-stopping jolt of energy.  The world freezes and fades to gray as you feel a sensation like your entire body being sucked through a two inch tube.  As suddenly as the sensation began, it stops.";
say "[line break]";
say "Type ABOUT for credits, notes, and hints.";
choose row 1 in Table of Basic Help Options;
change description entry to "Interface was conceived in about 1984 by a 14 year-old boy who wanted desperately to work for Infocom.  This boy programmed a parser and the Interface design using Commodore Basic at that time but, of course, shelved it a few years later when Infocom morphed into an unrecognizable creature.  That boy grew up to be me.  Now, in 2009, I ran into Inform 7 while dreaming nostalgically on the internet.[line break][line break]To be true to my childhood delusions of grandeur, I have coded up the game as an exercise in ‘what would it have been like’ and to surprise a few fellow-gamer friends of mine.  While working on the game I discovered the IF community and decided to enter it into the IF Comp after making a few minor tweaks in a vain attempt to bring the design a step or two closer to the modern era of IF.[line break][line break]Thus, you find before you a throw-back/relic delivered with care.  While I don’t pretend this to be a work of art, I do feel it’s an honest attempt of which my 14 year-old version of me would have approved; warts, plot holes, and all.  I hope the 1984 version of you, if applicable, enjoys as well.".

At 7:47 AM:
	say "Memories from after the horror in the chair attack you without warning.  You recall returning to consciousness to see Gilby and your Uncle again, but in a different way.  In fact, everything was different.  Your entire body felt numb.  The world sounded different, as if heard through a cheap dime store radio speaker.  Even the color of the world became faded and blotchy.  You saw relief wash over your Uncle Floyd's face, but he and Gilby immediately began to argue about tests, safety protocols, faulty equipment, and a bunch of other things you didn’t follow.  You felt a pang of instant terror when your Uncle began screaming about his nephew being interfaced into a backup robot unit.  Gilby promised to fix things.  He sputtered that he'd take ‘it’ home and spend all night fixing everything.  'Well, you'll have to, won't you!' your Uncle fumed just before storming out of the room in a frenzied huff.";
say "[line break]";
Say "Gilby reached toward you but hesitated for a moment.  He sneered, 'I don't get paid enough to deal with him!  I'm sorry little man, but tomorrow I'm going to go find a new job!  Someday I might try to figure out how to get you back into your own body… if I can find the time.'  Then, he flipped some sort of switch somewhere and everything faded again.  The next thing you knew, you found yourself here.  Everything still looks and feels wrong.  Now, all you want is to return your life back to normal...".

At 7:48 AM:
	say "Gilby steps around from behind you.  ‘OK, so that should do it.’ he says with a weary expression.  ‘Are you in there, little man?’ he asks rhetorically.  ‘I've been up all night... I gotta get some sleep.  Just hang out here until… Well, I got an interview later.  After that, I’ll take care of you.'  Then he looks at you very seriously.  ‘DON’T go hurting that robot you’re in.  I NEED those parts for my next job.  If your Uncle won’t give me what I need, I’ll find another sucker who will.  Oh, and don’t go doing anything stupid.  I need you to stay in there until I can… well… ‘fix’ you.’ The irony in his voice as he says this does not give you a good feeling that he is sincere about anything but his own motives.  And don’t go and try and leave, I’ve got you locked in.'  He displays a set of keys to prove his point and then slaps his hand tightly around them in a dramatic gesture to reinforce that he’s in control.

Gilby leaves by the north door, the keys to your freedom rattling in his hand.";

Section 2 - Living Room

Living Room is a room.  "This living room is rather plain, ordinary, unamusing, and otherwise your basic boring.  A worn couch rests beneath a set of closed curtains.  Opposite of the couch is a small, untidy desk.  From here you can see an entryway to the east and a large double-door frame, complete with missing double doors, on the north wall which leads into another room.".

There are some containers called the curtains.  "[one of]The curtains are, if possible, even more boring than the rest of the room.  They do not appear to have been opened for a very long time.  In fact, they may never have been opened.[or]The curtains remain forever closed against the outside world.[stopping]".

The Curtains are closed.
The Curtains are scenery.
The Curtains are in the Living Room.
Instead of opening the Curtains: say "[One of]The curtains probably opened at one time, but they have long since lost that ability.[or]Nope, its not happening.[stopping]".

There is a worn couch which is scenery in the Living Room.  "The couch is faded and worn.  It reminds you of garage sales that your mother would drag you to near the local college campus.".
Understand "Sofa" as Couch.
Instead of pushing the Couch: say "It won't budge."

There is a desk which is scenery in the Living Room.  "It looks like a K-Mart blue-light special and has seen better days.  An empty computer docking station rests in one corner.  An inspection of the front of the desk reveals a[if desk drawer is open]n open[otherwise] closed[end if] drawer.".
The desk is a supporter.
The desk drawer is closed.
The Desk Drawer is openable.
The Desk Drawer is not portable.
The Desk Drawer is Scenery.
The Desk Drawer is part of the desk.

There is a door frame in the living room.
The door frame is scenery.
Understand "doorframe" as door frame.
Understand "double-door frame" as door frame.
Understand "double door frame" as door frame.

There are things called papers in the desk drawer.
The description of the Papers are "There are a number of scratchings for various projects.  However, the majority of the papers seem to contain sketches of some sort of machine that is roughly pyramid-shaped with a rounded object at its peak.  The sketches show a tread system, much like a tank would employ, at the base of the pyramid.  As well, there are sketches for a pyramid-shaped contraption that  fits onto the machine but is removable.  There are many wires hooked to this contraption and below it are the words 'Transfer Mechanism.'".
Understand "sketches" as Papers.
[Instead of taking the papers: say "They are too fine for your crude graspers to manipulate."]
Understand "sketches" as papers.
Understand "sketch" as papers.
Understand "scratchings" as papers.
Understand "scratch" as papers.
Understand "scratching" as papers.
Understand "scraps" as papers.
Understand "scrap" as papers.
understand "drawings" as papers.
understand "drawing" as papers.

There is a computer docking station in the Living Room.
The Computer Docking Station is scenery.
The description of the Computer Docking Station is "It won't do you much good without a computer.".
Instead of taking Computer Docking Station: say "It's held in place by several wires.".
understand "dock" as docking station.

wires are part of the docking station.
wires are scenery.
Instead of taking the wires, say "They are securely attached to the docking station.".
Instead of cutting wires, say "[one of]Your weak grippers cannot apply enough force to cut through the wires.[or]Really, you can't.  You are not strong enough.[stopping]".

There are some dull pencils in the desk drawer.
The description of the Dull Pencils are "They look rather like, um, dull pencils."
Instead of taking pencils: say "That is yesterday's technology.  Besides, they are much too dull for a high-tech adventure such as this!"
Understand "pencil" as pencils.

There are candy wrappers in the desk drawer.
the description of Candy Wrappers are "Looks like 3 Musketeers, mostly.  The wrappers are quite devoid of candy."
Before taking Candy Wrappers for the first time: 
	say "What's the use?  The candy is already gone and you wouldn't be able to eat the candy if it was there!  Oh well...";
	continue the action.

There is a plastic card. 
The description of the Plastic Card is "The card is white and made of a flimsy plastic.  On one side, printed in brash lettering is 'Caliber Laboratories', the name of your Uncle's electronics company.  On the opposite side is a magnetic strip."
The Plastic Card is inside the Desk Drawer.
There is a magnetic strip.
The magnetic strip is part of the plastic card.

Section 3 - Entryway

The Entryway is east of the Living Room.  "You are in a small area that joins several portions of the house together.  To the south is a plain white front door whose only feature is an unusual looking lock.  A plain stairway leads up to a second floor.  Through a doorway to the east is the kitchen, to the west is the living room, and a thin hallway runs north.".

Upstairs is a room.  Upstairs is up from the Entryway.
There is a Plain Stairway.
The Stairway is a supporter.
The Plain Stairway is in the Entryway.
The Plain Stairway is scenery.
The description of the Plain Stairway is "The long flight of stairs is dull and ordinary.  There are about 10 steps leading up to the second floor.".

Instead of going up from the Entryway:
	if the plyboard is on the Plain Stairway begin;
		say "You move up, allowing the plyboard to be used as a make-shift ramp.  However, after about three steps you run out of board and must abort your attempt.  It was a very nice try, though.";
		Stop the action;
	otherwise;
		say "The treads that serve to move you around merely bump into the lowest step.  They are not large enough to grab and lift you up these steps.";
	end if.

Section 4 - Hallway

The Hallway is north of the Entryway.
The description of the hallway is "This hallway is the center of the house.  A few steps, er, tread rolls to the west is a rec room or you may go south to the entryway.  On the north wall is the door to the bathroom.".

Section 5 - Neglected Kitchen

There is a room called the Neglected Kitchen.
The description of the Neglected Kitchen is "This kitchen is certainly used, but not much effort is put forth to keep it clean.  The small space is crammed with a stove, refrigerator, sink, cabinet, and a table.  There is an open doorway to the north that leads to the dining room.  Also, there is a side door on the east wall.  The entryway lies to the west.[if water pool is in the Kitchen]  Centered in the doorway to the dining room to the north is a large dangerous looking pool of water.[end if]".
The Neglected Kitchen is east of the entryway.

There is a supporter called kitchen table.  "You don't really want to.  You're not sure it has been cleaned... ever!"
The Kitchen Table is not portable.
The Kitchen Table is scenery.
The Kitchen Table is in the Neglected Kitchen.

There is a steak knife.  "Lying forgotten on the table is a kitchen knife."
The description of the knife is "The knife has a single sharp blade with a wooden handle.  It appears to have slain many apples in its lifetime."
The knife is on the Kitchen Table.

There is a container called the kitchen cabinet in the Kitchen.
The kitchen cabinet is closed.
The kitchen cabinet is openable.
The kitchen cabinet is scenery.
Instead of opening the kitchen cabinet, say "It only contains 'just add water and eat' type meals, which are completely useless to you.  You close it again so that SOMETHING in this kitchen is kept tidy.".

There is a container called a refrigerator.
The refrigerator is in the kitchen.
The refrigerator is closed.
The refrigerator is openable.
The carrying capacity of the refrigerator is 25.
The refrigerator is scenery.
Understand "fridge" as refrigerator.
The description of the refrigerator is "The fridge is unremarkable in every way.[if the refrigerator is open]  The interior, however, could only be described as belonging to a bachelor.  It is bare except for a bottle of ketchup and a moldy thing that is no longer recognizable.[end if]".
Instead of opening the refrigerator when the refrigerator is closed:
	say "You open the refrigerator.";
	Now the refrigerator is open;
	try examining the refrigerator.

There is a stove.
The stove is scenery.
Instead of opening the stove, say "That's not such a hot idea.".
Understand "oven" as stove.
The stove is in the Kitchen.

There is a container called a kitchen sink.
The kitchen sink is scenery.
The carrying capacity of the kitchen sink is 7.
Instead of switching on the kitchen sink, say "Electricity and water don't mix, and last you looked, you were electrical!".
The kitchen sink is in the kitchen.

There is a Bottle of Ketchup.
The description of the Ketchup is "Everyone needs a bottle of ketchup!  This one looks like it has been well used as it is nearly empty.".
Instead of opening the Bottle of Ketchup, say "Your grippers do not have the strength to open the Bottle of Ketchup.".
The Bottle of Ketchup is in the refrigerator.
The ketchup is scenery.
Instead of doing something with the Ketchup, say "It’s arguable that ketchup goes good with anything, but in this case you really don’t need it.".

There is a moldy thing.
The description of the Moldy Thing is "It is something that was probably once food.  It looks as if it has been ignored for several months.  At the moment you are rather glad you can't smell anything as the stench of the Moldy Thing would probably be overpowering.".
The Moldy Thing is in the refrigerator.
The moldy thing is scenery.
Instead of doing something with the moldy thing, say "You can barely stand to look at it, let alone mess with it.  You find yourself actually being glad you can't smell anything at the moment.".

Section 5.4 Driveway (Part a)

There is a room called Driveway.  "The driveway continues to the street south of here, leading away to your freedom.  As well, you can see the front porch and yard of the decrepit house to the east.  A rusty fence acts as a barrier between the driveway and the yard.  Mounted atop the fence is a battered mailbox which you can reach from here."

Section 5.5 - Garage

There is an unlocked closed door called the side door.
[
Before going through the closed side door: 
    say "(first opening the side door)[command clarification break]"; 
    silently try opening the side door. 
]

There is a room called the Minuscule Garage.
The side door is east of the Neglected Kitchen and west of the Garage.
the Side Door is scenery.
The description of the Garage is "The garage is extremely small, barely big enough to house even a single compact car.  There are various forgotten odds and ends lying about.  The garage door lies [if garage door is open]open[else]closed[end if] to the south, held in place by the automatic opener.  You can see that Gilby has been here because the button to open and close the garage door has been hung on a nail high out of your reach.".
The Garage can be empty or full.
The Garage is full.
In the Garage is a sheet of plyboard.
Understand "board" as Sheet of Plyboard.
Understand "wood" as Sheet of Plyboard.
Understand "plywood" as plyboard.
Understand "wood" as plyboard.
Instead of putting plyboard on water pool:
	say "The board is too large to fit inside the doorway where the water pool is located.";
	stop the action.
Understand "puddle" as water pool.

There is a button in the garage.
The button is scenery.
The description of the button is "It resembles a doorbell button.  Normally it would be mounted at a reasonable height, but it appears Gilby really wants you to stick around as he has hung it on a nail high up on the wall, well out of your reach.".
Instead of doing something with the button, say "Gilby has moved it too high up for you to reach.".
Things can be Throw_Light, Throw_OK, or Throw_Heavy.
Things are usually Throw_Heavy.
The knife is Throw_OK.
The plastic card is throw_light.

Instead of throwing something (called the object) at something (called the target):
	if the object is throw_light begin;
		say "You watch as [the noun] flitters helplessly to the ground.  You will need a substantial projectile for a worthy toss.";
		Now the object is in the location of the player;
	end if;
	if the noun is throw_heavy begin;
		say "Like a bowling ball slipping from the fingers of a four-year old, you watch as [the object] thumps impotently to the floor.  Your weak extensions need something a bit lighter for a good pitch.";
		Now the object is in the location of the player;
	end if;
	if the noun is throw_ok begin;
		if the location of the player is the garage and the target is the button begin;
			if the garage is full begin;
				say "You are unable to get a good throwing angle at the button since the car takes up most of the floor space.  If the car was out of the way, you suspect you’d have no problem at all.";
			otherwise;
				if the garage door is closed begin;
					say "[The object] arcs gracefully and bounces off the button with a thump.  Instantly, the garage door rumbles open to reveal the driveway to the south.";
					if the button is _unawarded begin;
						award 5 points;
						Now the button is _awarded;
					end if;
					Now the object is in the location of the player;
					Now the garage door is open;
				otherwise;
					say "You toss [the object] at the button.  Again the button yields and the garage door resumes it's guard against intruders.";
					Now the object is in the location of the player;
					Now the garage door is closed;
				end if;
			end if;
		otherwise;
			say "You toss [the object] at [the target], but it merely bounces off and rattles to the ground.";
			Now the object is in the location of the player;
		end if;
	end if

In the garage is a container called a car.
The car is locked.
The car is not opaque.
The car has a carrying capacity 99.
The description of the car is "It's a little foreign job in disrepair.  You suspect there is more Bondo than actual metal from the looks of it.  It is a testament to the smallness of the garage that even a little foreign vehicle takes up most of the available space."
Instead of taking the car, say "It's far too heavy to carry." 
Instead of Entering the Car, say "You can't work out how to lift yourself into the car.  And, if you could, you appear to be the wrong shape to fit through the doorway and into the seat.".
Before locking keylessly the car while the car is open:
	say "What a piece of junk!  The door locks are busted.  The only way you can lock it is from the outside using the keys.";
	Stop the action.
Before locking the car with keys while the car is open:
	say "What a piece of junk!  The door locks are busted.  The only way you can lock it is from the outside using the keys.";
	Stop the action.

Before taking plyboard while the garage is full:
	say "There is not much room in the garage. The car, while a compact model, takes up so much space that your rather bulky frame cannot squeeze by it to retrieve the sheet of plyboard."; 
	stop the action.

There is a door called the garage door.  
The Garage Door is scenery.
The description of the Garage Door is "It's pretty much like any other garage door.  You open it, drive in or out, and close it again.".
The Garage Door is south of the Garage and north of Driveway.
The Garage Door is closed.
Instead of opening the garage door when the garage door is closed, say "You can't open it by hand... er... gripper.  The automatic opener has it pinned in place.".
Instead of closing the garage door when the garage door is open, say "You can't reach it at the moment.  Besides, the button worked better before.".

There is an automatic opener.  "The garage door opener appears to be in working order, but the button that controls it has been moved up the wall, well out of reach.".
The Automatic Opener is scenery.
The Automatic Opener is in the Garage.
Instead of doing something with the automatic opener, say "[one of]It's mounted far above you, too far to reach.[or]No, really.  It's way up there.[or]You haven't grown any since the last time.  It's out of reach.[stopping]".

Section 6 - Dining Room

There is a thing called a china cabinet. 
Understand "Cabinet" as China Cabinet. 
Understand "Cab" as China Cabinet. 
The China Cabinet can be Unmoved or Moved.
The China Cabinet is Unmoved.
The China Cabinet is scenery.
The description of the China Cabinet is "The cabinet looks very out of place in the dining room, and in the house as a whole, as it's the only nice piece of furniture in the place.  It is open faced and looks as if it was made by hand of sturdy oak.  While it was obviously intended to be a China Cabinet, it actually holds knick-knacks that only true old school geeks could love.  Oddly, there are scratches in the hardwood floor to the side of it."
The China Cabinet is in the Dining Room.

There are some knick-knacks.  "One man's treasures are your trash.  There is nothing useful among the knick-knacks.".
The knick-knacks are in the dining room.
The knick-knacks are scenery.
understand "knick knacks" as knick-knacks.
understand "knick" as knick-knacks.
understand "knicks" as knick-knacks.
Understand "knacks" as knick-knacks.
Understand "knack" as knick-knacks

The description of the Dining Room is "This tucked off to the side room serves as a Dining Room.  There is a small rickety looking table with chairs, and a formal china cabinet is on the east wall.  The cabinet seems a bit out of place with the rest of the household furnishings.[If China Cabinet is moved]  The China Cabinet has been shoved aside revealing a small alcove to the east.[end if]  The kitchen is off to the south.".

The Dining Room is north of the Neglected Kitchen.

There is a supporter called the dining room table.  
The description of the dining room table is "It's old and rather rickety.  It looks like something purchased for less than a dollar at a garage sale.".
The Dining Room Table is in the Dining Room.
The Dining Room Table is scenery.
There are chairs in the Dining Room.
the Chairs are supporters.
The chairs are scenery.
Understand "chair" as chairs.
Instead of pulling or pushing the chairs, say "You don't really relish dragging chairs about.  Besides, they belong at the table.".

There are scratches.
The scratches are in the dining room.
The description of the scratches are "On the floor next to the cabinet are several minor scratches.  They are long and straight implying that the cabinet has been moved back and forth several times.".
The scratches are scenery.

Instead of pushing the China Cabinet:
	If the China Cabinet is Unmoved begin;
		Change China Cabinet to Moved;
		say "The cabinet must be resting on some type of casters since your weak robotic body is able to easily push it.  You shove it back several feet, revealing a small alcove to the east."; 
		If the cabinet is _unawarded begin;
			Award 10 points;
			Now the cabinet is _awarded;
		end if;
	Otherwise;
		Change China Cabinet to Unmoved;
		say "You return the China Cabinet to its original position."; 
	end if.

Before going east from Dining Room:
	if China Cabinet is unmoved:
		Say "You can't go that way."; 
		Stop the action.

A water pool is a backdrop.  "A low spot in the hardwood floor has become a catching place for water.  While the source is not evident, it doesn't take much imagination to think that Gilby isn't too worried about a leak in the water lines, a pool of water on the floor, or anything else that doesn't have to do with electronics.  An able-bodied human could, albeit neglectfully, step right over this pool.  However, your robot body would be in great danger if you tried to plow through it."
Water pool is in the Neglected Kitchen.

Before going north from the Neglected Kitchen:
	if Water pool is in the Kitchen begin;
	say "You can't, the water pool is in the way and you'd short circuit for sure!";
	stop the action;
otherwise;
	Continue the action;
end if;

Instead of dropping the body towel on the water pool:
	Try rubbing the water pool.
Instead of putting the body towel on the water pool:
	Try rubbing the water pool.

There is a dirty wet towel.
The description of the Dirty Wet Towel is "The towel is still damp from cleaning up that dangerous water pool."
Before rubbing water in the presence of the water pool:
	if the player is holding the body towel begin;
		say "It's about time someone cleans up that mess.  It's not doing the hardwood floors any favors.  After soaking up the water, you push the now damp towel into the corner of the kitchen.";
		Award 10 points;
		Remove Body Towel from play; 
		Now the Dirty Wet Towel is in the kitchen;
		Remove Water pool from play;
		stop the action;
	otherwise;
		say "I think you need something with which to clean this mess!"; 
		stop the action;
	end if.

Instead of taking Dirty Wet towels:
	say "It is too wet.  You'd short circuit for sure!"

Section 6.5 - Hidden Alcove

There is a room called Hidden Alcove.  "This little nook is just large enough to hold you.  On the east wall, looking very out of place, is a strong looking gray metal door.  The door has no markings or handles.  Mounted into the wall next to the door is a small slit lined with black material.  The slit looks like it's designed to have something roughly the size of a credit card inserted into it.".
The Hidden Alcove is east of Dining Room.
The security door is a locked door.
The Security Door is scenery.
The Security Door is east of Hidden Alcove and west of Secret Lab.
The description of the Security Door is "The door is smooth and shiny.  Oddly enough, there is no handle or key.  A small slit in the wall next to the door is the only distinguishing feature of the portal.";
To say key-refusal for (locked-thing - Security Door):
     say "The security door has no visible locking mechanism.".

There is a container called a small slit. 
The description of the small slit is "The slit is built into the wall next to the door.  It is about 3 inches wide and you can just see a bit of metal inside implying a machine that will catch whatever is put into the slit.";
The Small Slit is part of the Security Door.
Understand "Slot" or "Small Slot" as Small Slit. 

Before inserting something into the small slit:
	If the noun is Plastic Card begin;
		Continue the action;
	Otherwise;
		Say "Try as you may, you can't fit [the Noun] into the thin slit."; 
		Stop the action;
	end if.

Instead of unlocking the Security Door with something:
	say "And how do you propose to do that?". 

Understand "slot" as Slit.

After inserting something into the small slit:
	Now the Security Door is unlocked;
	Now the Security Door is open;
	Say "Deep inside the heavy security door, tumblers tumble, catches release, and the door swings open, allowing you passage to the east.";
	Award 10 points. 

Section 7 - Rec Room

The Rec Room is north of the Living Room. 
The Rec Room is west of the Hallway.
The description of the Rec Room is "This shabby room seems to be where Gilby must spend his down time as there are various trinkets and time wasters scattered about.  Sitting prominently in the center of the room is a crumb filled easy chair that faces a high definition television and stereo.[Line Break][line break]Exits lead to the north, east, and south.".

There are trinkets.
Understand "time wasters" as trinkets.
Understand "junk" as trinkets.
Trinkets are scenery.
Trinkets are in the Rec Room.
The description of the trinkets is "[one of]They are things that only a juvenile in an adult’s body could love.[or]You don’t have time to waste with these time-wasters!  You have to put your life back in order![stopping]".
Instead of taking the trinkets, say "You won't need those."

There is a supporter called an easy chair.  
The description of the easy char is "The chair probably came from a thrift store.  It has the remains of many meals ground into its fabric.".
The easy chair is in the Rec room.
The easy chair is scenery.

There is a thing called crumbs.
The description of the crumbs is "These crumby crumbs are really crumbly."
The crumbs are on the easy chair.
The crumbs are scenery.

There is a device called a high definition television.  "It looks sharp!  The remote is nowhere in sight, but there is at least a large power button on the set.".
The television is in the Rec Room.
The television is scenery.
Understand "tv" as Television. 
Understand "tele" as Television. 
There is a power button.  The power button is part of the television.
Instead of pushing the power button:
	if the television is switched on begin;
		say "You press the button, and the set turns off.";
		Now the television is switched off;
	otherwise;
		say "You press the button, and the set begins to glow.";
		Now the television is switched on;
	end if.

At  8:00 AM: change the TVShowNumber to 2.
At  8:30 AM: change the TVShowNumber to 3.
At  9:00 AM: change the TVShowNumber to 4.
At  9:30 AM: change the TVShowNumber to 5.
At 10:00 AM: change the TVShowNumber to 6.
At 10:30 AM: change the TVShowNumber to 7.
At 11:00 AM: change the TVShowNumber to 8.
At 11:30 AM: change the TVShowNumber to 9.
At 12:00 PM: change the TVShowNumber to 10.
At 12:30 PM: change the TVShowNumber to 11.
At  1:00 PM: change the TVShowNumber to 12.
At  5:00 PM: change the TVShowNumber to 20.
At  6:00 PM: change the TVShowNumber to 22.
At  6:30 PM: change the TVShowNumber to 23.
At  7:00 PM: change the TVShowNumber to 24.
At 10:00 PM: change the TVShowNumber to 30.
At 10:30 PM: change the TVShowNumber to 31.
At 11:00 PM: change the TVShowNumber to 32.
At 11:30 PM: change the TVShowNumber to 33.

Every Turn:
	if player is in the Rec room and the television is switched on begin;
		say "On the TV you can see [TVShow in row TVShowNumber  of the Table of Television Shows].[line break]";
	end if;

Table of Television Shows
TVShow
"the genius is chasing the road runner"
"someone demonstrating the cutting edge technology of clapping on and clapping off"
"the Iron Chef is working with the mystery food of 'Komodo Dragon'"
"an infomercial showing the face of a woman who is ecstatic about her new rubber gloves"
"an overpaid group of friends whining about nothing in particular"
"a guy who liked the company so much, he bought it"
"an old Lone Ranger episode in crisp, clear high-def black and white"
"Maxwell Smart struggling with a phone booth and a shoe"
"the Skipper chasing Gilligan, again, and again, and again"
"a remarkably clean man doing a dirty job"
"an old episode of the xyzzy files"
"soap operas"
""
""
""
""
""
""
""
"a rerun of the A-Team, mohawk and all"
""
"the evening depression, er, news."
"the evening news"
"the movie of the week, Rocky 22"
"the movie of the week, Rocky 22"
"the movie of the week, Rocky 22"
"the movie of the week, Rocky 22"
"the movie of the week, Rocky 22"
"the movie of the week, Rocky 22"
"the nightly news"
"the local nightly news"
"late nigzht tv"
"late night tv"
"late night tv"
"late night tv"
"late night tv"

There is a device called a stereo.  "The stereo is definitely high dollar equipment.  You notice that the volume knob is turned up quite high.  If this thing was turned on, it would probably wake up the entire neighborhood.".
The Stereo is in the Rec room.
The Stereo is scenery.
Understand "Radio" as Stereo. 
Understand "Stereo" as Stereo. 

Instead of listening to the stereo:
	if the stereo is switched off begin;
		say "There is nothing to hear.  The stereo is turned off.";
	else;
		say "With the volume up so loud, it's hard not to listen to the stereo!";
	end if.

After switching on the stereo:
	Change stereo_count to 1;
	say "A nearly overpowering wave of loud rock music begins to pour out of the stereo speakers.  It is loud enough to wake the dead.".

Every Turn:
	If the stereo is switched on begin;
		change stereo_count to stereo_count plus 1;
		if stereo_count is 4, change stereo_count to 1;
		if stereo_count is 3 begin;
			If the player is in Outside_the_house begin;
				say "You continue to hear the rhythmic thumping of the stereo from within the house.";
			otherwise;		
				say "[one of]Thunderous music is blasting from the stereo.[or]The loud music continues to rattle the house.[or]Music keeps bumping from the stereo.[or]The stereo continues it's job at an ear-bleeding volume.[at random]";
			end if;
		end if;
	end if.

After taking keys for the first time:
	Award 20 points;
	Continue the action.

Section 8 - Bedroom

Bedroom is north of the Rec Room. 
The description of the Bedroom is "You find yourself in a room that only a bachelor would call a bedroom.  It is quite bare except for a simple mattress and box springs lying on the floor as a bed.  On the south wall, next to the door you entered through, is a full length mirror.";

There is a thing called a full length mirror.
The Full Length Mirror is scenery.
The Full Length Mirror is in the Bedroom.
The description of the Full Length Mirror is "You are gripped with shock as you peer into the mirror.  While you knew you were no longer in your body, nothing could prepare you for the cold reality of the reflection of a robot staring back at you.  Your chassis is roughly pyramid shaped with a white opaque globe mounted at the top, probably concealing the camera that is serving as your eyes.  Below the chassis is a system of short treads that allow you to move about.  An extension which ends in a simple gripper mechanism protrudes from each face of the pyramid.  You are struck by the shabbiness of the construction.  Around the edges of the various panels are small gaps and the occasional exposed wire underscoring the makeshift nature of your new body.".
Instead of searching the Full Length Mirror, try examining the Full Length Mirror.

There is a supporter called the bed.
The bed is scenery.
The bed is in the bedroom.
Understand "Mattress" as the bed.
Understand "Box Springs" as the bed.
The description of the bed is "It looks as if someone just dropped the box springs and mattress here, right on the floor, and called it good."
Instead of looking under the bed, say "You can't.  The bed is lying on the floor."

Section 9 - Hall Bathroom

The Hall Bathroom is a room.
The description of the Hall Bathroom is "The bathroom is no bigger than a generous closet.  A small cabinet with a single door, small sink, and a mirror mounted above it is just inside the door.  A toilet sits stuffed beneath an angled portion of the ceiling which probably accommodates the stairway.  Next to the toilet is an economy bathtub.".

There is a closed unlocked door called the bathroom door.
The Bathroom door is scenery.
The Bathroom door is north of the Hallway and south of the Hall Bathroom.
[
Before going through the closed bathroom door: 
    say "(first opening the bathroom door)[command clarification break]"; 
    silently try opening the bathroom door. 
]
There is a container called the cabinet.
The cabinet is scenery.
The cabinet is closed.
The cabinet is openable.
The body towel is in the cabinet.
The description of the body towel is "A hitchhikers best friend!".
The Cabinet is in the Hall Bathroom.
The body towel is throw_light.

There is a bathroom mirror.
The Bathroom Mirror is scenery.
The bathroom mirror is in the bathroom.
The description of the Bathroom Mirror is "It's quite cloudy, but you can just make out something that looks like a white upside-down fish bowl reflected on its surface.".
Instead of searching the bathroom mirror, try examining the bathroom mirror.

There is a container called a small sink.
The small sink is in the bathroom.
The small sink is scenery.
The carrying capacity of the small sink is 7.
Instead of switching on the small sink, say "Electricity and water don't mix, and last you looked, you were electrical!".
Understand "basin" as sink.
Understand "wash basin" as sink.
Understand "faucet" as sink.
Understand "tap" as sink.
Understand "taps" as sink.

There is a thing called a toilet in the bathroom.
The toilet is scenery.
Understand "John" as toilet.
Understand "throne" as toilet.
Understand "camode" as toilet.
Instead of entering the toilet, say "I think there is a concept or two about this story you haven't yet grasped.".
Instead of examining the toilet, say "You see nothing special about the toilet, although it could stand to be flushed.".

Flushing is an action applying to one thing.
Understand "flush [something]" as flushing.
Check flushing: 
    if the noun is not a toilet, say "That can't be flushed, you silly little human!." instead. 
Carry out flushing: 
    say "[One of]The toilet makes a satisfying 'whooshing' sound.  That was so fun, you're tempted to do it again![or]OK, that was more fun the first time.[or]Hmmm... it appears we are still in the northern hemisphere.[or]The toilet obliges.[stopping]".

Filling is an action applying to one thing.
Understand "fill [something]" as filling.
Check filling:
	if the noun is:
	-- sink: try switching on sink instead;
	-- bathtub: try switching on bathtub instead;
	-- otherwise: say "You would if you knew how, which you don't, so are aren't.".

There is a container called a bathtub.
The bathtub is in the bathroom.
The bathtub is scenery.
Instead of switching on the bathtub, say "Electricity and water don't mix, and last you looked, you were electrical!".
Instead of entering the bathtub, say "You lift your legs over the side of the tub and slip in... oh... wait... you don't have legs.  Nevermind.".
The carrying capacity of the bathtub is 20.
Understand "tub" as bathtub.
Understand "bath" as bathtub.

There is a device called the shower.
The shower is in the bathroom.
The shower is scenery.
Instead of switching on the shower, say "The realization that electricity and water don't mix shocks you out of your desire to turn the shower on.".

Understand "wash" as rubbing.

Section 10 - Porch

There is a closed lockable locked door called the front door.
The description of the front door is "The door is pretty plain, except that Gilby has put a two way lock on the door so that it can be locked from the inside as well as the outside.".
Understand "lock" as front door.
The Front Door is scenery.

The Front Door is south of Entryway and north of the Dilapidated Porch.

Instead of opening the locked front door:
	say "A freshly installed double-sided lock stops you.";
	stop the action.

The description of the porch is "This small porch has a shabby railing and peeling paint.  A short, wide set of three steps leads down to the south to the yard that is divided by a cracking sidewalk."

There is a supporter called a plain staircase.
The plain Staircase is in the Porch.
Understand "Stairs" as Staircase.
Understand "Stairs" as Stairway.
Understand "Steps" as Staircase.
Understand "Steps" as Stairway.
Understand "Stairway" as Staircase.
Understand "Staircase" as Stairway.
The plain staircase is scenery.
The description of the plain Staircase is "Three shallow wooden steps make up the stairs.  [if sheet of plyboard is on the plain staircase]A piece of plyboard has been laid across the steps allowing you safe passage up and down[otherwise]You doubt that the robot body you are trapped in will be able to make it down those steps[end if].".

After putting plyboard on the plain staircase for the first time:
	Award 10 points;
	Continue the action.

Before going south from Porch:
	if the Plyboard is not on the plain Staircase begin;
		say "Your body is not designed to go up or down stairs.  You can only roll along mostly flat surfaces.";
		Stop the action;
	end if;

Before going down from Porch:
	if Sheet of Plyboard is not on plain Staircase begin;
		say "Your body is not designed to go up or down stairs.  You can only roll along mostly flat surfaces.";
		Stop the action;
	end if;

There is a shabby railing.  
The description of the railing is "The railing is thin and bent, but serves its purpose.".
The shabby railing is scenery.
The shabby railing is in the porch.
Understand "rail" as railing.

There is peeling paint.
The description of the peeling paint is "Obviously little care has been taken to keep the exterior of the house in good shape.  The white paint is peeling in large sections.  Many flakes have fallen to the ground below where they remain.".
The peeling paint is scenery.
The peeling paint is in the porch.

Section 11 - Sidewalk

The Sidewalk is down from Porch.
The sidewalk is south from Porch.

The description of the sidewalk is "You are in the yard of the small house.  From here you can see the street is filled with nearly identical looking homes.  The sidewalk runs through the neglected front yard.  The grass and weeds are too thick to allow you to pass.  A rusty fence surrounds the yard.  Next to a gate on the south side of the yard is a battered mailbox.".

There is a locked closed door called the rusty gate.
The Rusty Gate is south of the sidewalk and north of the street.
The Rusty Gate is scenery.
the description of the Rusty Gate is "It's rusted shut.";

There is a container called a battered mailbox in the sidewalk.
The mailbox is fixed in place.
The mailbox is closed.
The mailbox is openable.
The mailbox is scenery.
The carrying capacity of the mailbox is 2. 

There is a package. 
The description of the package is 
"The plain brown package is about the size and shape of a brick, although you doubt that it actually contains a brick. Without your sense of touch you are really not sure how heavy the package is. A handwritten address label reveals that the package is for Gilby and has come from Caliber Labratories, your Uncle's business!".
The carrying capacity of the package is 2.
The package is closed.
The package is unopenable.
The package is in the mailbox.
Instead of opening the package:
	if the second noun is nothing begin;
		say "With your bare grippers?"; 
		stop the action;
	end if.
There is a thing called addressing.
Understand "address" as addressing.
Understand "addresses" as addressing.
Addressing is part of the package.
Addressing is scenery.
The description of the addressing is "The package was sent from Caliber Laboratories, your Uncle's business, to Gilbys home address.  The labels are in your Uncle's handwriting.".

Before unlocking:
	if noun is package begin;
		if the second noun is the knife begin;
			if the package is closed begin;
				say "The knife easily slips in and cuts the package open.  A few pieces of debris fall from the freshly cut opening."; 
				Now the package is open;
				Award 5 points;
				try searching the package;
				stop the action;
			otherwise;
				say "It's already been opened.";
				stop the action;
			end if;
		otherwise;
			say "I don't see how!"; 
			stop the action;
		end if;
	otherwise;
		continue the action;
	end if.

Instead of cutting while the player does not hold the knife:
	say "You don't have anything with which to cut!".	
Instead of cutting while the player holds the knife:
	if the noun is:
	-- Gilby:
		If Gilby is asleep and Gilby is on the bed and the player is in the bedroom begin;
			say "No!  Your mother is raising you right.  The guy is defensless while he's asleep.";
		otherwise;
	 		Say "Your body reacts much too slowly, and Gilby avoids you easily.  In fact, your not even sure he noticed."; 
		end if;
	-- Package: try unlocking package with knife;
	-- Moldy thing: say "You create a small incision, throwing spores into the air.  After a moment, they disappear from sight.";
	-- Grass: say "You, frankly, don't feel like doing Gilby the favor of mowing the lawn a blade of grass at a time.";
	-- Otherwise: Say "You only accomplish making light surface scratches.  Your cheap robot parts can't apply enough force to cut more deeply.".

Inside the package is a circuit board.
The description of the circuit board is "It looks just like any other circuit board you have seen.".
The circuit board is Throw_Light.
The package is throw_OK.
The note is throw_light.

The description of the note is "The note reads:[line break][line break]Gilby,[line break][line break]Enclosed is the rebuilt 9.73 x 42.6 Merlin board we discussed.  I’m still quite certain that this was the cause, dispite your theory.  I strongly suggest you try it out before this gets any uglier.[line break][line break]If you fail me in this, I will cut off your research dollars immediately.[line break][line break]Floyd[line break][line break]If you still had a face, the color would drain from it.  Your Uncle doesn’t realize that Gilby is defecting from Caliber Laboratories!".
The note is in the package.

There is a fence.
The fence is scenery.
The fence is in the sidewalk.

Before going south from the porch:
	Now the mailbox is in the sidewalk;
	Now the fence is in the sidewalk;
	Continue the action.

Before going down from the porch:
	Now the mailbox is in the sidewalk;
	Now the fence is in the sidewalk;
	Continue the action.

There is a front yard.
The description of the front yard is "Same as the house, the yard has been neglected.  There is no decorative touches and it hasn’t seen a lawn mower in quite some time.".
The front yard is in the sidewalk.
The front yard is scenery.

There is some grass.
The description of the grass is "It’s been a while since anyone mowed this lawn.".
The grass is part of the front yard.
Understand "lawn" as grass.

There are some weeds.
The weeds are part of the front yard.

Section 11.1 Driveway

Before going south from garage:
	Now the mailbox is in the driveway;
	Now the fence is in the driveway;
	Continue the action.

Before going south from the Driveway:
	say "A tickle in your brain... er... CPU insists that if you leave now, you'll likely never return.  You may be free of the house, but you'll be stuck in this stupid robot body.  Are you sure you wish to make a break for it now (Yes or No)?";
	if player consents begin;
		continue the action;
	otherwise; 
		say "You were tempted, but... no.  You're going to stick this out!";
		stop the action;
	end if;

Section 11.2 Freedom

There is a room called Freedom.
Freedom is south of Driveway.

After going to freedom:
	say "After only a moment of hesitation you flee the house to find refuge.";
	Award 15 points;
	say "[line break]";
	say "The next days and months are difficult for you.  You attempt a reunion with your parents, but find it hard to connect with them given your new bodies’ limitations.  They simply regard you as one of the spoiled neighbor kid's toys.  Even sophisticated tricks, such as writing messages that recall shared memories, do not force them to believe you are anything other than the result of a cruel joke meant to antagonize them in their time of mourning over their missing son.";
	say "[line break]";
	say "In desperation, you find your way to Uncle Floyd’s house.  You have to lay in ambush outside his front gate.  Upon seeing you, he bursts into uncontrollable sobbing and brings you into his home to live with him.  A few years of this pseudo-life go by before, ironically, Floyd is able to obtain some equipment from a competing company whose now famous head engineer is, of course, Gilby.  Using the turncoat’s equipment, Uncle Floyd is able to, at least, transfer you to some upgraded equipment that allows you to better communicate and connect with the people you care about.  However, you never again experience the world around you as only a true human can.";
	say "[line break]";
	say "Having nothing but time to kill, you often reflect on your life and decide that if you had it to do all over again that you would not have run away from Gilby’s little shack of a house.  Rather, you would have sought out the solution that would have returned your life to normal!";
	End the game saying "You won!  Or did you lose?  You are not sure!  If only you had it all to do over again...".

Section 12 - Secret Lab

The description of the Secret Lab is "The appearance of this room stands in sharp contrast to the rest of the house.  There are mountains of gleaming new humming electronics wired together.  You quickly surmise from the numerous Caliber Laboratories property tags on the equipment that much of it has come from your Uncle’s company and probably without his permission.  Gilby must spend the majority of his time here tinkering with these impressive gadgets.[line break][line break]In one corner of the lab is a simple table holding what you recognize to be your body!  After transferring your brain to this robot, Gilby must have slumped your now limp body here for safe keeping.[line break][line break]One particularly complex machine stands out from the rest as it is much larger than the others and has a group of wires leading to your body.".

There are property tags in the Secret Lab.
The property tags are scenery.
The description of the property tags are "The tags are found on various pieces of equipment and are used to indicate that the tagged item is property of Caliber Laboratories.".
Understand "tag" as property tags.
Instead of taking the property tags, say "They are permanently affixed to their given piece of equipment and cannot be removed without specialized tools.".

There is a lab table in the Secret Lab.
The Lab Table is scenery.
On the Lab Table is your body.
The description of your body is "Your body lies limp, but seems no worse for wear.".
Your body is scenery.
instead of putting [something] on the lab table, say "The table is much too crowded.".

There is some complex machinery. 
The description of the Complex Machinery is "The Complex Machine is rather large compared against the other electronics in the room.  Leading from the back of the machine is a bundle of wires that are attached about the head of your body that lies on the table.  On the face of the Complex Machine is a rectangular port[if port contains the circuit board] that has the replacement circuit board plugged into it[otherwise] that seems to be missing some sort of equipment[end if].  There is also a large conspicuous looking lever with labels above and below it as well as an important looking switch.  Attached to the side opposite of your body is another bundle of wires that lead to a harness.";
The Complex Machinery is in the Secret Lab.
The Complex Machinery is scenery.
Understand "machine" as machinery.

There is a rectangular port.
The description of the port is "[if the port contains the circuit board]The circuit board has been snugly mounted into the port.[otherwise]The rectangular port appears to be designed to hold a circuit board of some type.[end if]".
The rectangular port is part of the complex machinery.
The rectangular port is a container.

There is a pyramid shaped harness.
The description of the harness is "The harness is made out of thick leather and is molded into a rough pyramid shape approximately the size of your chassis.  There is a bundle of wires running from the harness to the Complex Machine.  The harness looks to the same size and hape of your chassis."
The pyramid shaped harness is in the secret lab.
The pyramid shaped harness is wearable.
Before going west from the secret lab:
	if player is wearing the harness begin;
		say "The harness you are wearing is attached to the complex machine. While the length of the cable is generous, leaving the room would be out of the question."; 
		stop the action;
	end if;
	if player is holding the harness begin;
		say "The harness you are holding is attached to the complex machine. While the length of the cable is generous, leaving the room would be out of the question."; 
		stop the action;
	end if.


There is a thing called a bundle of wires.
The bundle of wires is in the secret lab.
The bundle of wires is scenery.
Instead of pushing or pulling the bundle of wires, say "The bundle of wires is firmly attached to the machinery.".
Understand "cable" as bundle of wires.

Before inserting something into the rectangular port:
	if the noun is circuit board begin;
		say "It slips snugly into place."; 
		Now the circuit board is inside the port;
		stop the action;
	otherwise;
		say "[one of]It doesn't fit properly.[or]It's the wrong shape.[at random]"; 
		stop the action;
	end if.

There is a lever. 
The lever can be Pushed Up or Pushed Down.
The lever is Pushed up.
The lever is part of the Complex Machinery.
The description of the lever is "The lever looks as haphazardly put together as the rest of the machine. There are two pieces of masking tape stuck to the metal above and below the lever. The top piece of tape is peeling a bit, and has 'H to R' written on it in ink that had to have come from a Sharpie. The bottom piece has 'R to H'. The lever is currently in the [if lever is pushed up]up[otherwise]down[end if] position.". 

instead of pushing or pulling the lever:
	if the lever is pushed down begin;
		say "The lever is now in the up position, under the 'H to R' tag."; 
		Now the lever is pushed up;
	otherwise;
		say "The lever is now in the down position, above the 'R to H' tag."; 
		Now the lever is pushed down;
	end if.

Flipping is an action applying to one thing.
Understand "flip [something]" as flipping.
Check flipping:
	say "You amuse yourself by attempting to flip it for a moment, and then give up.".

Instead of flipping the important looking switch:
	try pushing the important looking switch.

Instead of switching on the important looking switch:
	 try pushing the important looking switch.
Instead of pulling the important looking switch:
	 try pushing the important looking switch.

There is a device called an important looking switch.
The Important Looking Switch is part of the Complex Machinery.
The description of the switch is "The switch is unlabeled, but is painted red.  The area around the switch is painted with yellow and black stripes.  The overall effect is that something big will happen if the switch is pressed.".
After pushing the Important Looking Switch:
	if the circuit board is inside the rectangular port begin;
		if the player is wearing the harness begin;
			if lever is pushed up begin;
				say "Your body, which was lying still on the table, suddenly begins to shudder as the machinery starts pulling a brain imprint from the now blank mind.  You watch, too late, as the machine gathers up the empty imprint and then begins the process of transferring, through the harness, to you.  Your last thought is about the terrible mistake you have made as the new, blank imprint is pushed into the robots body, forcing you out and into oblivion.[line break]";
				End the game in death;
			otherwise;
				Say "You watch in wonder as the machinery sends a pulse of energy through the wire and into the harness which begins to shake violently.  The world turns blue, and you forget your name and how many arms you have.  A splitting headache guides your return to consciousness.  Instinctively, you raise your hand to your head in a vain attempt to sooth it against the pain.  You are startled to feel some wires taped to your forehead.  Your recent adventure inside the body of the robot bursts back into your thoughts.[line break][line break]";
				Say "Opening your eyes you are not surprised to find yourself lying on the table in the secret lab.  Elated to be in your own body once again, you quickly strip off the wiring and rise.  After the massive headache complains about the sudden movement, you begin again a bit more slowly.
Across the room you spy the sad little homemade robot chassis in which you were recently trapped.  You feel the urge to give it a good kick, but decide against it as it would probably only cause you more pain.  With the full employ of a human body once again available, you quickly find your way out of the house.[line break][line break]";
				Say "Several squad cars squall to a stop in front of Gilby’s house.  Your Uncle, accompanied by several police officers, pop out of the vehicles.  He runs to you and wraps you in a ferocious hug.  You are not sure whether to return his gesture or bite him in retribution.[line break][line break]";
				Say "He explains that once he calmed down he grew concerned.  Leaving you to be fixed by Gilby was probably the biggest air-head move he had made in his life.  So, along with the circuit board and note he had also placed a ‘Fly on the Wall’ in the package he sent to Gilby.  It was small enough that he knew it would go unnoticed by whoever opened the package.  Through the eyes of the Fly he saw Gilby had abandoned you in the robot.  He got the police involved to arrest Gilby, citing the stolen equipment in the secret lab.[line break][line break]";
				Say "Eventually you forgive your Uncle.  However, you and he have a standing agreement that all new inventions are to be demonstrated to you by someone else from now on.";
				Award 25 points;
				End the game in Victory;
			end if;
		otherwise;
			if the lever is pushed up begin;
				Say "You body, lying on the table, begins to shudder with energy that pulses through the wiring of the complex machine.  The brainless mass shakes violently as the machinery tries to pull its brain imprint yet again.  After a moment, your body gives up the useless fight, dying.[line break][line break]All hope of ever returning to your body died along with it.[line break]";
			otherwise;
				Say "The harness begins to jump around on its wires, as if some great energy is passing through it.  However, you are soon distracted as your body that is lying on the table soon shudders as well.  Shortly after the shuddering ceases, your body rises on its own.  Your body looks blankly ahead, as if there was no brain guiding its actions, and then collapses to the floor.  Although you cannot take a pulse, your are quite certain that your body is now lifeless.[line break][line break]All hope of ever returning to your body died along with it.[line break]";
			end if;
			End the game in death;
		end if;
	otherwise;
		say "The switch makes an impressive clicking sound, but nothing else seems to happen."; 
	end if;
	stop the action.

Section 14 - The Player

Instead of examining the player, say "You cannot see much of your own body as whatever is serving as your eyes is limited in its movements.  You assume yourself to be about 4 feet in height based on what is at eye level with you.  You can, however, see that your body is made of metal, some of which is certainly scrap originally intended for other uses.  The portions that you can see are basically symmetrical forming a boxy four-sided pyramid .  Each of the four faces of the pyramid has an arm-like extension with a gripper mechanism at the end.  These grippers are small and not very flexible, but allow simple manipulations.".
Understand "My Body" as Your Body. 
Understand "chassis" as yourself.
There is an extension.
The extension is scenery.
The description of the extension is "You take a moment to study the extensions and find that they, more or less, can serve as arms and hands to pick things up and use them in simple ways.".
The extension is part of the player.
Understand "extensions" as extension.
Understand "grippers" as extension.
Understand "gripper" as extension.
Understand "mechanism" or "mechanisms" as extension.
Understand "arm" or "arms" as extension.
Understand "bowl" as the yourself.
Understand "globe" as the yourself.
Understand "fish bowl" as the yourself.
Understand "tread" or "treads" as yourself.
Understand "panel" or "panels" as yourself.
Understand "gap" or "gaps" as yourself.

Instead of tasting, say "You cannot find a mouth anywhere on your body."
Instead of Smelling, say "It appears you have no nostrils!".
Instead of Touching, say "Touch pads are not standard equipment on this new body of yours.".
Instead of eating, say "First you'll have to find your mouth.".
Instead of drinking, say "First you'll have to find your mouth.".
Instead of sleeping, say "You don't seem to be able to relax in the human sense.  Your processors just keep processing against your will."
Instead of climbing, say "Your grippers are too weak to hold your weight, and treads make it worse.".
Instead of thinking, say "Your processors crunch numbers for a bit.".
Instead of looking under automatic opener, say "You see the rest of the garage."
Instead of looking under the harness, say "You have seen all there is to see of the harness."
Instead of looking under something, say "You find that you can't bend over to take a look.".

The carrying capacity of the Player is 4.

Section 15 - Gilby

Gilby is a Targeted Patroller.
The description of Gilby is "Gilby, your Uncle Floyd's (former) assistant, looks like a classic Hollywood stereotypical nerd.  His round face and ugly mustache are impossible to ignore.[If Gilby is on the bed]  It appears that Gilby had a late night because he is sacked out while fully dressed, probably still wearing yesterday's clothes.[end if][If Gilby is on the bed and Gilby is carrying the keys]  You can see a set of keys half-covered by his motionless hand.[end if]".

Gilby is on the bed.
Gilby can be asleep, awake, Heading for Shower, Showering, finding keys, leaving, or gone.
Gilby is asleep.
Gilby has a number called grumpiness.
The grumpiness of the Gilby is 0.
Gilby has a number called DelayCounter.
The DelayCounter of Gilby is 0.
Understand "man" as Gilby.
Understand "guy" as Gilby.
Understand "assistant" as Gilby.
Understand "hands" as Gilby.

A lousy mustache is part of Gilby.
The description of the lousy mustache is "You stare at the mustache for a while until you get the uneasy feeling that it's staring back at you.".
Instead of taking the mustache, say "He seems rather attached to it.".

There are things called some keys.
Understand "keyring" as keys.
Understand "ring" as keys.
Gilby carries the keys.
The description of the keys are "There are several keys on the keyring.".
The keys unlock the Front Door.
The keys unlock the car.
The keys are Throw_OK.

Every turn:
	If the player is in the rec room for the first time begin;
		if SnoringClueGiven is 0 begin;
			say "You hear an odd noise coming from the north.  Is that... snoring?";
			Change SnoringClueGiven to 1;
		end if;
	end if;


Every turn:
	If Gilby is asleep and Gilby is on the bed and the player is in the bedroom begin;
		choose a random row in the Table of Snoring Remarks;;
		Say "[One of]You hear loud snoring, although it sounds rather odd through the cheap microphone that serves as your 'ears.'[or][Remark entry].[stopping]";
	end if;

Table of Snoring Remarks
Remark
"Gilby continues his nearly endless drone of snoring"
"More snoring emanates from Gilby"
"The seemingly ceaseless sawing of logs continues from the bed"
"Gilby snorts and sputters and then continues his well-practiced snore."

Before taking keys:
	if Gilby has the keys and Gilby is on the bed begin;
		say "You attempt to take the keys from Gilby, but as you tug at them, he begins to move. He snatches the keys away, eyes you wearily, and shuffles a bit in the bed."; 
		Now the grumpiness of Gilby is the grumpiness of Gilby plus 1;
		stop the action;
	otherwise; 
		continue the action;
	end if.


Instead of taking keys while Gilby is carrying the keys:
	say "[One of]He pushes you away when you make the attempt.[or]Stop it!.[stopping]";
	Now the grumpiness of Gilby is the grumpiness of Gilby plus 1.


Every Turn:
	If the grumpiness of Gilby is three for the first time:
		 say "Gilby eyes you with obvious disdain.  'If you don’t stop bothering me, I’ll shut you down… permanently!’ he snarls.[If Gilby is on the bed]  He flops his head back onto the pillows.[end if]";
		Now the grumpiness of Gilby is the grumpiness of Gilby plus 1.


Every turn:
	if the stereo is switched on and Gilby is asleep begin;
		if Gilby is on the bed begin;
			if Gilby is carrying the keys begin;
				Now the keys are on the bed;
			end if;
			Now Gilby is in the Bedroom;
		end if;
		Now Gilby is Targeted;
		Now the drive of Gilby is 100;
		Now the Destination of Gilby is the Rec Room;
		Activate Gilby;
	end if;

Every turn:
	if Gilby is in the Rec room and Gilby is asleep begin;
		if the DelayCounter of Gilby is 0 begin;
			Now the DelayCounter of Gilby is the DelayCounter of Gilby plus 1;
		otherwise;
			Now Gilby is awake;
			Now Gilby is aimless;
			Now the OpeningCapability of the Gilby is UnlockedOnly;
			Now the ReclosingCapability of the Gilby is Reinstate;
			Now the Drive of Gilby is 100;
			Activate Gilby;
			Now the stereo is switched off;
			Now the DelayCounter of Gilby is 0;
			trHeadsForShower in 20 turns from now;
			if the player is in the Rec room begin;
				say "Gilby switches the stereo off mumbling rude things about you and your Uncle Floyd in the process. He glares at you for a moment and sighs deeply.  Then, with a start, his eyes grow wide with a memory and he utters 'Ohm'gawd... the interview!' and begins the ritual dance of getting ready for the day.";
			otherwise;
				say "The house shaking music coming from the stereo is suddenly gone.";
			end if;
		end if;
	end if.

At the time when trHeadsForShower:
	Now Gilby is Heading for Shower;
	Now the drive of Gilby is 95.;
	Now Gilby is Targeted;
	Now the Destination of Gilby is the Hall Bathroom;
	Activate Gilby.

Every Turn:
	if Gilby is in the Hall Bathroom and the player is in the bathroom begin;
		Say "Gilby eyes you with disgust.  'Get outta here!' he bursts as he shoves you back out of the tiny bathroom.  He closes the door, blocking your way back in.";
		Now the player is in the Hallway;
	end if;
	if Gilby is Heading for Shower and Gilby is in the Hall Bathroom and player is in the hallway and the bathroom door is open begin;
		say "Gilby gives you a hard awkward stare.  He slowly closes the bathroom door, seemingly unpracticed at the need for privacy in the shower.";
	end if;
	if Gilby is Showering and Gilby is in the Hall Bathroom and player is in the hallway begin;
		Now the Bathroom door is closed;
		Now the bathroom door is locked;	
		Say "You can hear the sounds of the shower coming from behind the bathroom door.";
	end if;
	If the location of Gilby is the bathroom and Gilby is heading for shower begin;
		Now Gilby is Showering;
		trLooksForKeys in 15 turns from now;
	end if;

At the time when trLooksForKeys:
	Now Gilby is Finding Keys;
	Now the Bathroom Door is unlocked.

Every turn:
	if Gilby is Finding Keys and Gilby is not carrying keys begin;
		Now Gilby is Targeted;
		Now the DelayCounter of Gilby is 0;
		Now the bathroom door is open;
		Now the destination of Gilby is the location of the keys;
		Now the drive of Gilby is 100;
		Activate Gilby;
	end if;
	If Gilby is Finding Keys and Gilby is in the location of the keys begin;
		If the player is carrying the keys begin;
			Say "Gilby startles you by putting his angry face two inches away from your electronic eye!  'Gimme back my keys, you stupid robot!' he sneers and he tears they keys from your grippers.  'I am outta here!' he snarls.";
			Now the Gilby is in the Location of the Player;
		otherwise;
			if the Location of the Keys is the Location of the Player begin;
				Say "Gilby is here searching for his keys.  Finding them, he picks them up and puts them in his pocket.  'I am outta here!' he snarls.";
				Now Gilby is in the Location of the Player;
			otherwise;
				Now Gilby is in the Location of the keys;
			end if;
		end if;
		Now Gilby is carrying the keys;
		Now Gilby is leaving;
		Now Gilby is Targeted;
		Now the ReclosingCapability of the Gilby is leave;
		Now the Destination of Gilby is the Garage;
		Activate Gilby;
	end if;


Every Turn:
	if Gilby is in the Garage and Gilby is leaving begin;
		If the DelayCounter of Gilby is 0 begin;
			Now the DelayCounter of Gilby is the DelayCounter of Gilby plus 1;
		otherwise;
			If the player is in the garage begin;
				Say "Seeing you, Gilby presses some sort of button somewhere on your chassis.  It appears he has disabled you as you can no longer move at all.  All you can do is simply watch as Gilby jumps into his car.  Soon the garage door opens, the car backs out, and the garage closes once again.  After several long moments, you regain the ability to move about.";
			else;
				Say "A deep rumbling noise is coming from the direction of the garage.  It stops for a moment, then begins again.  After a couple of seconds it stops and remains quiet.";
			end if;
			Remove the car from play;
			Now the garage is empty;
			Now Gilby is gone;
			Remove Gilby from play;
		end if;
	end if

Instead of waking Gilby while Gilby is on the bed:
	 say "You shake and prod him for a bit.  After a few moments he groggily raises his head, gives you an exasperated expression, and rolls back over.";
	Now the grumpiness of Gilby is the grumpiness of Gilby plus 1.

Instead of waking Gilby while Gilby is not on the bed:
	say "But... but... but he's already awake!".

Every turn:
	If the grumpiness of Gilby is greater than four begin;
		Now Gilby is awake;
		say "'‘That’s it!  I’ve had enough of you.’ Gilby says with obvious disdain.  ‘I don’t care if you are his nephew.  I’m gonna get another job anyway.’ He mutters under his breath as he reaches out for one of your panels.  He looks directly into your camera and sneers ‘Nightie nite!’ as he flips a switch, cutting all power to your body.  Although you cannot feel a thing, and are not aware of it, he continues a shutdown sequence which wipes your imprint from the machine’s memory and from existence itself.";
		End the game in death;
	end if.

Instead of pulling Gilby while Gilby is on the bed, try waking Gilby.
Instead of pushing Gilby while Gilby is on the bed, try waking Gilby.

To say Gilby Doing Something:
	if Gilby is awake begin;
		change AwakeDescrNumber to AwakeDescrNumber + 1;
		if AwakeDescrNumber > the number of rows in Table of Awake Actions, Change AwakeDescrNumber to 1;
		say "[ActionText in row AwakeDescrNumber  of the Table of Awake Actions]";
	end if;
	if Gilby is heading for shower begin;
		say "heading to take a shower";
	end if;
	if Gilby is leaving begin;
		say "jingling his car keys, whistling, looking sharp, and in a hurry";
	end if;
	if Gilby is asleep begin;
		say "looking bleary eyed and agitated";
	end if;
	if Gilby is finding keys begin;
		say "looking for his keys";
	end if.

To report arrival of (Bod - a patroller) up from (place - a room) while Gilby is not asleep:
	say "[The Bod] climbs up the stairs from below, [Gilby doing something]."
To report arrival of (Bod - a patroller) down from (place - a room) while Gilby is not asleep:
	say "[The Bod] hops runs down the stairs, [Gilby doing something]."
To report arrival of (Bod - a patroller) coming (way - a direction) from (place - a room):
	say "[The Bod] arrives from [the way], [Gilby doing something]."
To report departure of (Bod - a patroller) going (way - a direction) to (place - a room):
	say "[The Bod] heads [way], [Gilby doing something]."

Table of Awake Actions
ActionText
"yawning"
"pulling off his shirt"
"making rude snorting sounds to clear his head"
"eating a power bar"
"searching for a clean spoon"
"sipping on a cup of coffee"
"mumbling about how mornings come too early in the day"
"scratching at his dandruff"
"looking for clothes that aren't too dirty"
"checking his e-mail on his mobile phone"
"sniffing his armpits and considering whether a shower is really required"
"looking to see if he still owns a comb"
"trying to find a matching sock"
"looking for his wallet"
"rubbing his eyes"
"giving practice answers to interviews"
"staring off in deep thought"
"looking at his watch with alarm"
"desperately trying to find his shoes"
"looking for the 'directions I printed off last night"

rule for printing room description details of Gilby while Gilby is heading for shower:
	say " wearing a bath robe"

rule for printing room description details of Gilby while Gilby is asleep:
	if Gilby is on the bed begin;
		say " sleeping deeply.  There is something under one of his hands";
	otherwise;
		say " looking bleary eyed";
	end if.

Section 16 - End of Game and Scoring

The maximum score is 90.

Table 1 - Rankings
Score	Rank
0	"feeble minded" 
15	"in brain freeze" 
30	"lost in your own thoughts" 
45	"artificial intelligence" 
60	"brain power" 
70	"positronic brain!" 
90	"master mind!" 

[
Scoring:
25 Transfer back to body & Win
10 Open secret door
10 Push China Cabinet out of way
10 Clean up water
10 Open Package with Knife
10 Use board to get down stairs
20 Distract Jordon with stereo so you can get key
]

Section 17 - Command Overrides, Action Definitions, General Action Overrides

Sitting is an action applying to one visible thing and requiring light. 
Understand "Sit on [something]" as sitting.
Understand "Sit in [something]" as sitting.
Instead of sitting [ something]:
	say "It would be impressive if you could figure out how to make your robot chassis sit.".

Instead of jumping:
	say "You'd probably break a tread if you tried it!".

Instead of singing:
	say "You have no external speakers with which to produce noise such as singing.".

Talking is an action applying to nothing.
Understand "talk" or "shout" or "yell" as talking.
carry out talking:
	say "You have no external speakers with which to communicate!".

Talking to is an action applying to one thing.
Understand "talk to [something]" or "talk at [something]" or "shout to [something]" or "shout at [something]" or "yell to [something]" or "yell at [something]" as talking to.

Check talking to:
    if the noun is not a person, say "Do you expect a response?" instead. 

carry out talking to:
	say "Without an external speaker [noun] has no chance of hearing you!".

XYZZYing is an action applying to nothing.
Understand "XYZZY" as XYZZYing.
Carry out XYZZYing:
	say "Almost magically, you digest the ASCII characters XYZZY, break them into their binary counterparts, run them through several algorithms, and find that the only fruits of your analysis is the phrase: 'Nothing Happens.'".

Rule for printing a parser error when parser error is can't see any such thing:
    say "Your camera does not see that at the moment." instead.

Rule for printing a parser error when parser error is aren't holding that:
    say "You check, and you're not holding that in your graspers." instead.

Rule for printing a parser error when parser error is not a verb I recognise:
    say "That verb will be too hard to pull off while trapped in this robotic body." instead.

Rule for printing a parser error when parser error is can't see it at the moment:
    say "Your camera does not see that the moment." instead.

Rule for printing a parser error when parser error is didn't understand the way that finished:
    say "You lost me in the end there." instead.

Rule for printing a parser error when parser error is not enough of those available:
    say "There aren't enough." instead.

Rule for printing a parser error when parser error is noun did not make sense in that context:
    say "Sorry, your noun does not compute when used that way." instead.

Rule for printing a parser error when parser error is I beg your pardon:
    say "You are a robot of action.  Please input a little more data to go on!" instead. 

Kicking is an action applying to one thing. Understand "kick 
[something]" as kicking. 
Hitting is an action applying to one thing. Understand "hit [something]" 
as hitting. 

Going somewhere is Gross_Motor.
Pushing something is Gross_Motor.
Pulling something is Gross_Motor.
Kicking something is Gross_Motor.
Hitting something is Gross_Motor.
Attacking something is Gross_Motor.
Waking something is Gross_Motor.
Hitting something is Gross_Motor.
Sitting is Gross_Motor.
Looking Under is Gross_Motor.

Taking [something] is Fine_Motor.
Dropping [something] is Fine_Motor.
Cutting [something] is Fine_Motor.
Opening [something] is Fine_Motor.
Closing [something] is Fine_Motor.
Switching on [something] is Fine_Motor.
Switching off [something] is Fine_Motor.

Instead of Gross_Motor for the first time:
	say "You are stunned to find that, even though you are clearly thinking of a movement, your body seems to only jerk and twist in startling ways.  The realization that you are trying to control muscles that simply are not there washes through you with a chill.".

Instead of Gross_Motor for the second time:
	say "Again asking your new body to move, you succeed in only jerking forward about an inch.".

Instead of Gross_Motor for the third time: 
	say "After some experimentation, you find that ‘thinking’ thoughts such as pushing with your legs jerks you forward.  ‘Thinking’ of moving your arms causes the mechanical extensions of your new body to twitch.  With effort, you begin to define what thoughts cause what reactions, and can move clumsily about.";
	continue the action.

Instead of Gross_Motor for the fourth time: 
	say "You continue to move in starts and fits like a first-time driver behind a manual clutch.  However, you are getting smoother movement as you continue your attempts.";
	continue the action.

Before of Gross_Motor for the fifth time:
	say "Movement is coming much more naturally to you now, although you do have to give some conscious effort.";
	continue the action.

Before of Gross_Motor for the tenth time:
	say "You surprise yourself with the realization that you are moving about without having to think about it anymore.";
	continue the action.

Before Fine_Motor for the first time:
	say "The notion of simple small movements turn out to be a surprise.  You find that that normal thought of moving your arms and hands does not cause the expected reaction in your robotic body.  With your thought, the four extensions that serve as arms for this robotic chassis merely quiver.  The graspers at the end of the extensions only flex weakly.  It takes you several moments of concentration and experimentation to find that you must imagine pushing and pulling in new ways to cause the metallic arms to move.  Each arm takes a unique combination of push/pull notions to move independently.  Likewise, the graspers require unusual sensations of flexing your missing fingers to operate.  The effort is tedious and uncomfortable.  However, you will yourself to continue, if only to find a way out of this cursed machine.".

Instead of taking the water, say "I don't see how.  Your grippers have nothing with which to hold water!".

A thing can be light or heavy.
A thing is usually light.
The couch is heavy.
The desk is heavy.
The bed is heavy.
The kitchen table is heavy.
The Dining Room table is heavy.
The chair is heavy.
The mirror is heavy.
The car is heavy.
Gilby is heavy.
Instead of taking something that is heavy:
	say "The second-hand nature of the materials used to create your extensions and graspers would never hold up to the weight of [the noun].[line break]".
Instead of pushing something that is heavy, say "It's much too heavy to move.".
Instead of pulling something that is heavy,	say "It's much too heavy to move.".

Instead of taking something that is fixed in place:
	say "Your weak extensions can't possibly take that!".
Instead of pushing something that is fixed in place, say "It doesn't budge.".
Instead of pulling something that is heavy,	say "It doesn't budge.".

Instead of cutting yourself, say "Suicide is never the answer, little trooper.".
Instead of hitting yourself, say "Self loathing is not going to help you here.".
Instead of hitting [something], say "You'd probably break your extensions!".

Section 18 - Help

Table of Basic Help Options (continued)
title	subtable	description 
"Hints"	Table of Hints	--
"Credits/Contact"	--	"If you have enjoyed the game, or have any comments, bug fixes, etc., please drop me a line at bvegiard@activeinfosys.com.  These e-mails are my payment for the work I have done with this game, so please don't hesitate to click 'send'[line break][line break]This game would not be what it is without the beta testers, who are (alphabetically) Martin Bays, Richard Bos, Jeremy Freese, Wesley Osam, Poster, Brandon Scheuerman, Jason Travis, and Ramona White.[line break][line break]A big thank you goes out to Michael Callaghan for his Patrollers extension and, of course, Emily Short for several extensions.  Also, I'd like to thank the IF community, especially Jeremy Freese, for their assistance with technical challenges of the Inform 7 tool." 
"Settings"	Table of Setting Options	-- 


Table of Hints
title	subtable	description	toggle 
"How do I get out of the house?"	Table of Out of House Hints	""	hint toggle rule 
"How do I wake Gilby up?"	Table of Waking Guy Hints	""	hint toggle rule 
"How do I get rid of the water pool?"	Table of Water Pool Hints	""	hint toggle rule 
"Can I go up and down the stairs?"	Table of Stair Hints	""	hint toggle rule
"What can I do upstairs?"	Table of Upstairs Hints	""	hint toggle rule
"How do I get the rusty gate open?"	Table of Gate Hints	""	hint toggle rule
"What do I do outside?"	Table of Outside Hints	""	hint toggle rule
"How do I break the table?"	Table of Break_Table Hints	""	hint toggle rule
"How do I get the package open?"	Table of Package Hints	""	hint toggle rule
"OK, the package is open.  Now what do I do with this stuff?"	Table of Package_Stuff Hints	""	hint toggle rule
"Where can I use the stuff in the package?"	Table of Package_Stuff_Where_Use Hints	""	hint toggle rule
"How do I get past the security Door?"	Table of Security_Door Hints	""	hint toggle rule
"Is there a way to disarm the alarm and call off the cops?"	Table of Alarm Hints	""	hint toggle rule
"Whassup with the thing on the lab table?"	Table of Lab_Table Hints	""	hint toggle rule
"I'm stuck in the lab.  What do I do now?"	Table of Lab Hints	""	hint toggle rule
"How can I revive the other victims?"	Table of Victim Hints	""	hint toggle rule
"How do I return the doll to the girl?"	Table of Doll Hints	""	hint toggle rule
"That was a great ending to the game!"	Table of Great_Ending Hints	""	hint toggle rule
"That was a dumb ending to the game!"	Table of Dumb_Ending Hints	""	hint toggle rule

Table of Out of House Hints
hint	used 
"There are two exits: the front door, and the garage."	a number 
"The garage is tricky to open and the front door requires keys."	a number 
"Gilby has some keys.  Perhaps you could ask him to give them to you."	a number 
"...but you'd have to be able to talk.  There is a way to get them, though."	a number 
"While he's still asleep, surprise him somehow.  He'll drop the keys."	a number 
"Try TURN ON STEREO and see what happens."	a number
"Don't take too long to use the keys.  Gilby will want them back in a while"	a number
"If Gilby has left with the car, you can open the Garage."	a number
"The button to open the garage is too high, but there is a way to press it."	a number
"Try throwing things at the button."	a number
"The knife and the towel are not too heavy or too light to throw accurately."	

Table of Water Pool Hints
hint	used 
"Who's afraid of a little water pool?"	a number 
"Oh, wait... you are!  You're electrical... electricity and water, etc.  Um..."	a number 
"You need to get rid of it without touching it directly.  Have you tried the quicker picker upper?"	a number 
"Oh, yeah, no paper towels in the kitchen.  Are there any other towels?"	a number 
"There is a body towel in the bathroom."	a number 
"CLEAN UP WATER WITH TOWEL"

Table of Waking Guy Hints
hint	used 
"He sure sleeps soundly.  He only gets annoyed with you if you try to wake him, push him, or pull him."	a number 
"So, make him uncomfortable in a BIG way!"	a number 
"It's hard to sleep at a loud party."	a number 
"Loud parties have lots of loud music."	a number 
"Go to the Rec room and TURN ON STEREO."

Table of Stair Hints
hint	used
"Yes, kinda."	a number
"Not all stairs... only a small number of steps."	a number
"How does someone who only can use wheels (or treads) go up and down stairs?"	a number
"If you could build some sort of ramp.  You'll need some wood, though."	a number
"The plyboard is in the garage.  There is only enough to go up or down a few steps."	a number
"The front porch only has a few steps."	a number
"Go to the Front Porch and PUT PLYBOARD ON STAIRS"

Table of Upstairs Hints
hint	used
"Lots of things!"	a number
"...if you could only get there."	a number
"Which you can't.  Forget about it."

Table of Gate Hints
hint	used
"Be careful!  The street can be a dangerous place!"	a number
"Some oil would help."	a number
"Most people keep oil in the garage."	a number
"...except for this guy.  There's no oil on the shelves.  Have you looked under the car for drips?"	a number
"No oil there either, eh?  You have to really WANT to open the gate."	a number
"Too bad that even then you won't be able to.  Forget it."

Table of Outside Hints
hint	used
"Perhaps take up people watching?"	a number
"No people.  Have you gotten the mail?"	a number
"Hmmm... maybe there is something good in the package!"

Table of Break_Table Hints
hint	used
"Be violent!"	a number
"OK, now try being passive!"	a number
"Why would you want to break a (nearly) perfectly good table?  Forget it!"

Table of Package Hints
hint	used
"Be violent!"	a number
"No, really;  Be violent!  This is your only chance in this game to mutilate something!"	a number
"My kingdom for a letter opener."	a number
"...or a knife?"	a number
"CUT PACKAGE WITH KNIFE"

Table of Package_Stuff Hints
hint	used
"Read the note that is enclosed.  It should give you some clues."	a number
"You can use the circuit board somewhere in the house!  Start searching."	a number
"Using the circuit board is a big step toward getting out of this robot body!"

Table of Package_Stuff_Where_Use Hints
hint	used
"Somewhere inside the house."	a number
"But it's hidden..."	a number
"...behind something that looks out of place."	a number
"...in the dining room, PUSH CHINA CABINET"	a number

Table of Security_Door Hints
hint	used
"Put something appropriate in the slit."	a number
"Like a security card..."	a number
"...that you can find in the drawer of the desk in the living room."	a number

Table of Alarm Hints
hint	used
"Examine the control panel closely."	a number
"Perhaps pulling something out of it will help..."	a number
"...like the loose wire?"	a number
"What alarm?  There's no alarm in this game!"

Table of Lab_Table Hints
hint	used
"Did you examine it?"	a number
"It should look familiar."	a number
"That's your human body, silly!"	a number
"Now try and get back into it."

Table of Lab Hints
hint	used
"Look at the complex machinery closely"	a number
"The circuit board fits in the port"	a number
"The R to H on the lever stands for Robot to Human.  Try setting it to R to H!"	a number
"Putting the harness on will allow that machine to pull you out of the robot body!"	a number
"If you have all that accomplished, pushing the important looking button should make your day!"

Table of Victim Hints
hint	used
"Did you try putting her on the lab table?"	a number
"A good jolt of electricity should do the trick!"	a number
"Examine your chassis closely for a shocking surprise"	a number
"You really should stop reading hints for things that aren't in this game, y'know!"

Table of Doll Hints
hint	used
"This puzzle would tug at your heartstrings..."	a number
"...if it were a real puzzle."	a number
"But it's not, so forget it.  There's no doll in this game."

Table of Great_Ending Hints
hint	used
"Thanks!"	a number
"Since you enjoyed the game, please drop me an e-mail.  See the ABOUT screen for details."	a number
"That's all I have to say about that."

Table of Dumb_Ending Hints
hint	used
"Oh... sorry."	a number
"Perhaps I'll do better next time?"	a number
"Hey, if you're this upset, why don't you try writing your own game!"

Understand "key" as keys.

Outside_the_house is a region.  The Garage, porch, and sidewalk are in the Outside_the_house.

Walkthroughing is an action applying to nothing.
Understand "Walkthrough" as Walkthroughing.
Understand "Walk through" as Walkthroughing.
Understand "Walkthru" as Walkthroughing.
Understand "Walk thru" as Walkthroughing.
Carry out Walkthroughing: 
	say "A walkthrough has been provided in your game package.".

GilbyGoing is an action applying to nothing.
Understand "GilbyGo" as GilbyGoing.
Carry out GilbyGoing:
	Now the garage is empty.

Understand "something" as keys.

instead of closing the open security door, say "You don't see an easy way to accomplish that.".
instead of asking, say "You try, but find that you have no mouth whith which to speak!".
Instead of searching the China Cabinet, try examining the China Cabinet.

Rule for deciding whether all includes scenery: it does not.
